1.8 Průměrová rovina

```
> restart;
> with(LinearAlgebra):
> with(plots):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Rovnice kvadriky:

```
> Kv:=sort(expand(X.K.Transpose(X)),[x,y,z])=0;
```

$$Kv := a(1,1)\,x^2 + 2\,a(1,2)\,x\,y + 2\,a(1,3)\,x\,z + a(2,2)\,y^2 + 2\,a(2,3)\,y\,z + a(3,3)\,z^2 + 2\,a(1,4)\,x + 2\,a(2,4)\,y + 2\,a(3,4)\,z + a(4,4) = 0$$

Parametrické rovnice tečny s neasymptotickým směrem $sm = [u, v, w]$ a s bodem dotyku $M = [m, n, p]$

```
> sm:=[u,v,w]; M:=[m,n,p];
```

$$sm := [u, v, w]$$

$$M := [m, n, p]$$

```
> Tecna:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Tecna := [x = m + t\,u, y = n + t\,v, z = p + t\,w]$$

```
> Kv1:=simplify(eval(Kv,Tecna));
```

$$Kv1 := 2\,a(1,2)\,t^2\,u\,v + 2\,a(1,3)\,m\,t\,w + 2\,a(1,3)\,t\,u\,p + 2\,a(1,3)\,t^2\,u\,w + 2\,a(2,2)\,n\,t\,v + 2\,a(2,3)\,n\,t\,w + 2\,a(2,3)\,t\,v\,p + 2\,a(2,3)\,t^2\,v\,w + 2\,a(3,3)\,p\,t\,w + 2\,a(1,1)\,m\,t\,u + 2\,a(1,2)\,m\,t\,v + 2\,a(1,2)\,t\,u\,n + a(1,1)\,m^2 + a(2,2)\,n^2 + a(3,3)\,p^2 + 2\,a(1,4)\,m + 2\,a(2,4)\,n + 2\,a(3,4)\,p + a(1,1)\,t^2\,u^2 + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p + a(2,2)\,t^2\,v^2 + 2\,a(2,3)\,n\,p + a(3,3)\,t^2\,w^2 + 2\,a(1,4)\,t\,u + 2\,a(2,4)\,t\,v + 2\,a(3,4)\,t\,w + a(4,4) = 0$$

```
> A:=coeff(lhs(Kv1),t^2);
```

$$A := 2\,a(1,2)\,u\,v + 2\,a(1,3)\,u\,w + 2\,a(2,3)\,v\,w + a(1,1)\,u^2 + a(2,2)\,v^2 + a(3,3)\,w^2$$

```
> B:=1/2*coeff(lhs(Kv1),t);
```

$$B := a(1,3)\,m\,w + a(1,3)\,u\,p + a(2,2)\,n\,v + a(2,3)\,n\,w + a(2,3)\,v\,p + a(3,3)\,p\,w + a(1,1)\,m\,u + a(1,2)\,m\,v + a(1,2)\,u\,n + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w$$

```
> C:=sort(coeff(lhs(Kv1),t,0),[r,s,u]);
```

$$C := a(1,1)\,m^2 + a(2,2)\,n^2 + a(3,3)\,p^2 + 2\,a(1,4)\,m + 2\,a(2,4)\,n + 2\,a(3,4)\,p + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p + 2\,a(2,3)\,n\,p + a(4,4)$$

```
> B:=collect(B,[u,v,w]);
```

$$B := (a(1,3)\,p + a(1,2)\,n + a(1,4) + a(1,1)\,m)\,u + (a(2,2)\,n + a(2,3)\,p + a(2,4) + a(1,2)\,m)\,v + (a(1,3)\,m + a(3,3)\,p + a(2,3)\,n + a(3,4))\,w$$

```
> n1:=sort(coeff(B,u),[m,n,p]); n2:=sort(coeff(B,v),[m,n,p]);
  n3:=sort(coeff(B,w),[m,n,p]);
```

$$n1 := a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4)$$

$$n2 := a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4)$$

$$n3 := a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4)$$

```
> BD:=linalg[dotprod](sm,[n1,n2,n3],orthogonal);
```

$$BD := (a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4))\,u + (a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4))\,v + (a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4))\,w$$

```
> BD:=collect(BD,[m,n,p]);
```

$$BD := (a(1,1)\,u + a(1,2)\,v + a(1,3)\,w)\,m + (a(1,2)\,u + a(2,2)\,v + a(2,3)\,w)\,n + (a(1,3)\,u + a(2,3)\,v + a(3,3)\,w)\,p + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w$$

```
> ra:=sort(coeff(BD,m),[u,v,w]); rb:=sort(coeff(BD,n),[u,v,w]);
  rc:=sort(coeff(BD,p),[u,v,w]);
  rd:=sort(coeff(coeff(coeff(BD,m,0),n,0),p,0),[u,v,w]);
```

$$ra := a(1,1)\,u + a(1,2)\,v + a(1,3)\,w$$

$$rb := a(1,2)\,u + a(2,2)\,v + a(2,3)\,w$$

$$rc := a(1,3)\,u + a(2,3)\,v + a(3,3)\,w$$

$$rd := a(1,4)\,u + a(2,4)\,v + a(3,4)\,w$$

```
> PrumerR:=ra*x+rb*y+rc*z+rd=0;
```

$$PrumerR := (a(1,1)\,u + a(1,2)\,v + a(1,3)\,w)\,x + (a(1,2)\,u + a(2,2)\,v + a(2,3)\,w)\,y + (a(1,3)\,u + a(2,3)\,v + a(3,3)\,w)\,z + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w = 0$$

```
> U:=Vector[row]([sm[1],sm[2],sm[3],0]);
```

$$U := [u, v, w, 0]$$

```
> PrumerR:=collect(expand(U.K.Transpose(X)),[x,y,z])=0;
```

$$PrumerR := (a(1,1)\,u + a(1,2)\,v + a(1,3)\,w)\,x + (a(1,2)\,u + a(2,2)\,v + a(2,3)\,w)\,y + (a(1,3)\,u + a(2,3)\,v + a(3,3)\,w)\,z + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w = 0$$

Příklad: Průměrová rovina kvadriky $x^2 + 4\,y^2 + 16\,z^2 - 144 = 0$ sdružená se směrem $u = (-4, 2, 1)$.

```
> kv:=x^2+4*y^2+16*z^2-144=0;
```

$$kv := x^2 + 4\,y^2 + 16\,z^2 - 144 = 0$$

```
> a(1,1):=coeff(lhs(kv),x^2); a(2,2):=coeff(lhs(kv),y^2);
  a(3,3):=coeff(lhs(kv),z^2);
  a(1,2):=1/2*coeff(coeff(lhs(kv),x),y);
  a(1,3):=1/2*coeff(coeff(lhs(kv),x),z);
  a(1,4):=1/2*coeff(coeff(coeff(lhs(kv),x,1),y,0),z,0);
  a(2,3):=1/2*coeff(coeff(lhs(kv),y,1),z);
```

```
a(2,4):=1/2*coeff(coeff(coeff(lhs(kv),y,1),x,0),z,0);
a(3,4):=1/2*coeff(coeff(coeff(lhs(kv),z,1),y,0),x,0);
a(4,4):=coeff(coeff(coeff(lhs(kv),x,0),y,0),z,0);
```

$$a(1,1) := 1$$
$$a(2,2) := 4$$
$$a(3,3) := 16$$
$$a(1,2) := 0$$
$$a(1,3) := 0$$
$$a(1,4) := 0$$
$$a(2,3) := 0$$
$$a(2,4) := 0$$
$$a(3,4) := 0$$
$$a(4,4) := -144$$

```
> K;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 4 & 0 & 0 \\ 0 & 0 & 16 & 0 \\ 0 & 0 & 0 & -144 \end{bmatrix}$$

```
> PrumerR;
```

$$u\,x + 4\,v\,y + 16\,w\,z = 0$$

```
> Smer1:=[-4,2,1];
```

$$Smer1 := [-4, 2, 1]$$

```
> PrumerR1:=eval(PrumerR,[u=Smer1[1],v=Smer1[2],w=Smer1[3]]);
```

$$PrumerR1 := -4\,x + 8\,y + 16\,z = 0$$

```
> PrumerR1_ex:=solve(PrumerR1,z);
```

$$PrumerR1\_ex := \frac{x}{4} - \frac{y}{2}$$

```
> res1:=solve({kv,PrumerR1},{y,z});
```

$$res1 := \{y = \frac{x}{2} - \frac{1}{2}\,\mathrm{RootOf}(x^2 - x\,\_Z + \_Z^2 - 72, label = \_L2),$$
$$z = \frac{1}{4}\,\mathrm{RootOf}(x^2 - x\,\_Z + \_Z^2 - 72, label = \_L2)\}$$

```
> Kuz:=allvalues(res1);
```

$$Kuz := \{y = \frac{x}{4} - \frac{\sqrt{-3\,x^2 + 288}}{4}, z = \frac{x}{8} + \frac{\sqrt{-3\,x^2 + 288}}{8}\},$$
$$\{y = \frac{x}{4} + \frac{\sqrt{-3\,x^2 + 288}}{4}, z = \frac{x}{8} - \frac{\sqrt{-3\,x^2 + 288}}{8}\}$$

```
> Kuz1:=eval([x,y,z],Kuz[1]); Kuz2:=eval([x,y,z],Kuz[2]);
```

$$Kuz1 := \left[x, \frac{x}{4} - \frac{\sqrt{-3\,x^2 + 288}}{4}, \frac{x}{8} + \frac{\sqrt{-3\,x^2 + 288}}{8}\right]$$

$$Kuz2 := \left[x, \frac{x}{4} + \frac{\sqrt{-3\,x^2 + 288}}{4}, \frac{x}{8} - \frac{\sqrt{-3\,x^2 + 288}}{8} \right]$$

```
> T1:=evalf(eval(Kuz1,x=-9)); T2:=evalf(eval(Kuz1,x=5));
  T3:=evalf(eval(Kuz2,x=8)); T4:=evalf(eval(Kuz2,x=3));
```

$$T1 := [-9., -3.927050983, -0.2864745085]$$

$$T2 := [5., -2.398629880, 2.449314940]$$

$$T3 := [8., 4.449489743, -0.224744871]$$

$$T4 := [3., 4.788873605, -1.644436802]$$

```
> tec1:=evalm(T1+t*Smer1); tec2:=evalm(T2+t*Smer1);
  tec3:=evalm(T3+t*Smer1); tec4:=evalm(T4+t*Smer1);
```

$$tec1 := [-9. - 4\,t, -3.927050983 + 2\,t, -0.2864745085 + t]$$

$$tec2 := [5. - 4\,t, -2.398629880 + 2\,t, 2.449314940 + t]$$

$$tec3 := [8. - 4\,t, 4.449489743 + 2\,t, -0.224744871 + t]$$

$$tec4 := [3. - 4\,t, 4.788873605 + 2\,t, -1.644436802 + t]$$

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> tec1g:=plot3d(tec1,t=-3..3,s=-1..1,thickness=2):
  tec2g:=plot3d(tec2,t=-3..3,s=0..1,thickness=2):
  tec3g:=plot3d(tec3,t=-3..3,s=0..1,thickness=2):
  tec4g:=plot3d(tec4,t=-3..3,s=0..1,thickness=2):
> kvg0:=plots[implicitplot3d](kv,x=-15..15,y=-10..10,z=-10..10,gri
  d=[30,30,30],style=patchcontour,color=red,lightmodel=light1,scal
  ing=constrained):
> kvg:=plots[implicitplot3d](kv,x=-15..15,y=-10..10,z=-10..10,axes
  =frame,color=COLOR(RGB,250/255,250/255,250/255),style=patchconto
  ur,grid=[30,30,40],light=[90,-5,1,1,1],tickmarks=[3,3,3],orienta
  tion=[52,63],scaling=constrained):
> PrumerRov:=contourplot3d(PrumerR1_ex,x=-15..15,y=-10..10,grid=[2
  ,2],contours=40,color=COLOR(RGB,50/255,50/255,50/255),filled=fal
  se):
> PrumerRov0:=implicitplot3d(PrumerR1,x=-15..15,y=-10..10,z=-10..1
  0,style=patchnogrid,grid=[20,20,20],color=grey):
> Smerg:=plottools[arrow]([-5, -5, 12], vector(3*Smer1), 0.5, 1,
  0.2, cylindrical_arrow,color=black):
> Kuz1g:=spacecurve(Kuz1,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
  Kuz2g:=spacecurve(Kuz2,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
> display(kvg,PrumerRov,Smerg,Kuz1g,Kuz2g,tec1g,tec2g,tec3g,tec4g,
  axes=frame,orientation=[-130,60]);
```




```
> kvgW:=plots[implicitplot3d](kv,x=-15..15,y=-10..10,z=-10..10,axe
  s=frame,color=COLOR(RGB,150/255,150/255,150/255),style=wireframe
  ,grid=[20,20,40],light=[100,-20,1,1,1],tickmarks=[3,3,3],orienta
  tion=[52,63],scaling=constrained):
> St:=[1,2,eval(PrumerR1_ex,[x=1,y=2])]; Tet:=evalm(St+t*Smer1);
  M1:=eval(Tet,solve(eval(kv,[x=Tet[1],y=Tet[2],z=Tet[3]]),{t})[1]
  );
  M2:=eval(Tet,solve(eval(kv,[x=Tet[1],y=Tet[2],z=Tet[3]]),{t})[2]
  );
```

$$St := \left[1, 2, \frac{-3}{4}\right]$$

$$Tet := \left[1 - 4\,t, 2 + 2\,t, -\frac{3}{4} + t\right]$$

$$M1 := \left[1 - \frac{\sqrt{354}}{3}, 2 + \frac{\sqrt{354}}{6}, -\frac{3}{4} + \frac{\sqrt{354}}{12}\right]$$

$$M2 := \left[1 + \frac{\sqrt{354}}{3}, 2 - \frac{\sqrt{354}}{6}, -\frac{3}{4} - \frac{\sqrt{354}}{12}\right]$$

```
> Tetiva:=plot3d(evalm(M1+t*(M2-M1)),t=0..1,s=0..1,thickness=4):
> M11:=eval([x,y,z],[x=M1[1],y=M1[2],z=M1[3]]);
```

$$M11 := \left[1 - \frac{\sqrt{354}}{3}, 2 + \frac{\sqrt{354}}{6}, -\frac{3}{4} + \frac{\sqrt{354}}{12}\right]$$

```
> M22:=eval([x,y,z],[x=M2[1],y=M2[2],z=M2[3]]);
```

$$M22 := \left[1 + \frac{\sqrt{354}}{3}, 2 - \frac{\sqrt{354}}{6}, -\frac{3}{4} - \frac{\sqrt{354}}{12}\right]$$

```
> M1g:=plottools[sphere](M11,0.3):
  M2g:=plottools[sphere](M22,0.3): Stg:=plottools[sphere](St,0.3):
> Kuz1g:=spacecurve(Kuz1,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
  Kuz2g:=spacecurve(Kuz2,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
> Smerg:=plottools[arrow]([-5, 4, 4], vector(Smer1), 0.5, 1, 0.2,
  cylindrical_arrow,color=black):
> PrumerRov:=contourplot3d(PrumerR1_ex,x=-15..15,y=-10..10,grid=[2
  ,2],contours=40,color=COLOR(RGB,50/255,50/255,50/255),filled=fal
  se):
> display(Stg,M1g,M2g,Tetiva,kvg,PrumerRov,Kuz1g,Smerg,Kuz2g,axes=
  frame,orientation=[-131,66],view=[-15..12,-1.5..7,-5..7],scaling
  =constrained,light=[90,-5,1,1,1]);
```



Příklad: Průměrová rovina válcové plochy $x^2 + y^2 - 144 = 0$ sdružená se směrem $u = (1, 5, 2)$.

```
> kv2:=x^2+y^2-144=0;
```

$$kv2 := x^2 + y^2 - 144 = 0$$

```
> a(1,1):=coeff(lhs(kv2),x^2); a(2,2):=coeff(lhs(kv2),y^2);
  a(3,3):=coeff(lhs(kv2),z^2);
  a(1,2):=1/2*coeff(coeff(lhs(kv2),x),y);
  a(1,3):=1/2*coeff(coeff(lhs(kv2),x),z);
  a(1,4):=1/2*coeff(coeff(coeff(lhs(kv2),x,1),y,0),z,0);
  a(2,3):=1/2*coeff(coeff(lhs(kv2),y,1),z);
  a(2,4):=1/2*coeff(coeff(coeff(lhs(kv2),y,1),x,0),z,0);
  a(3,4):=1/2*coeff(coeff(coeff(lhs(kv2),z,1),y,0),x,0);
  a(4,4):=coeff(coeff(coeff(lhs(kv2),x,0),y,0),z,0);
```

$$a(1, 1) := 1$$

$$a(2, 2) := 1$$

$$a(3, 3) := 0$$

$$a(1,2) := 0$$
$$a(1,3) := 0$$
$$a(1,4) := 0$$
$$a(2,3) := 0$$
$$a(2,4) := 0$$
$$a(3,4) := 0$$
$$a(4,4) := -144$$

```
> K;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -144 \end{bmatrix}$$

```
> PrumerR;
```

$$u\,x + v\,y = 0$$

```
> Smer2:=[1,5,2];
```

$$Smer2 := [1, 5, 2]$$

```
> PrumerR2:=eval(PrumerR,[u=Smer2[1],v=Smer2[2],w=Smer2[3]]);
```

$$PrumerR2 := x + 5\,y = 0$$

```
> res1:=solve({kv2,PrumerR2},{x,y});
```

$$res1 := \{x = -30\ \mathrm{RootOf}(13\ \_Z^2 - 2, label = \_L5),\ y = 6\ \mathrm{RootOf}(13\ \_Z^2 - 2, label = \_L5)\}$$

```
> Kuz:=allvalues(res1);
```

$$Kuz := \{x = -\frac{30\sqrt{26}}{13}, y = \frac{6\sqrt{26}}{13}\}, \{x = \frac{30\sqrt{26}}{13}, y = -\frac{6\sqrt{26}}{13}\}$$

```
> Kuz1:=eval([x,y,z],Kuz[1]); Kuz2:=eval([x,y,z],Kuz[2]);
```

$$Kuz1 := \left[-\frac{30\sqrt{26}}{13}, \frac{6\sqrt{26}}{13}, z\right]$$

$$Kuz2 := \left[\frac{30\sqrt{26}}{13}, -\frac{6\sqrt{26}}{13}, z\right]$$

```
> kvg:=plots[implicitplot3d](kv2,x=-15..15,y=-15..15,z=-10..10,gri
  d=[30,30,30],style=patchcontour,color=red,lightmodel=light1,scal
  ing=constrained):
> PrumerRg:=implicitplot3d(PrumerR2,x=-15..15,y=-15..15,z=-10..10,
  grid=[2,2,2],contours=40,color=blue):
> ug:=plottools[arrow]([-5, -5, 10], vector(Smer2), .6, .8, .3,
  cylindrical_arrow,color=green):
> Kuz1g:=spacecurve(Kuz1,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
  Kuz2g:=spacecurve(Kuz2,x=-15..15,thickness=3,numpoints=1000,colo
  r=yellow):
> display(kvg,PrumerRg,ug,Kuz1g,Kuz2g,axes=boxed);
```



> v

> v